

---

**[ballyalley] New 300 baud tool**

---

From Paul Thacker via groups.io <lennier\_of\_the\_minbari@yahoo.com@groups.io>

Date Sat 8/21/2021 3:24 PM

To ballyalley@groups.io <ballyalley@groups.io>

I was getting a lot of gibberish out of tapes that Allen recently sent me, so I did a search to see if there were any alternate tools available for the 300 baud tapes using the Kansas City Standard. I found a program published fairly recently, TapeTool2 by Topten Software.

<https://www.toptensoftware.com/tapetool/>

After playing around with the options a bit, I got much better results for two tapes that were giving me trouble. I also like that audio filter support is built directly into the program. Both highpass and bandpass filters work well, bandpass maybe a bit better. I also found I sometimes needed to turn the noise threshold way down to make it sensitive to small signals. An example command for decoding to a binary file is:

```
tapetool2.exe ww6-raw.wav bandPass kansas.audioToBytes --noiseThreshold:0 ww6-bandPass-noise0.bin
```

Then the .bin file can be converted to a .txt file with prg2txt, or back to a good .wav file as usual.

I haven't tested it yet, but here's the missing Brain Bender from W&W Software Tape #6:

```
10 :return ;clear ;NT=0;BC=rnd (32)mul 8;FC=BC+4+rnd (32)mul 8
20 S=0;M=-1;&(20)=25;&(19)=32
21 if I=2clear ;goto 30
22 input "INSTRUCTIONS? 1=YES 2=NO"
23 if I=1goto 5000
25 clear
30 for Z=1to 9;T=rnd (2);if T=1@(Z)=1
40 if T=2@(Z)=0
70 S=S+@(Z);next Z;if S=0goto 30
95 U=1;M=M+1
100 for Y=-25to 25step 25
110 for X=-25to 25step 25
140 goto 2000
145 U=U+1
150 next X;next Y
160 gosub 600
170 A=0;B=0;box A,B,5,5,3
180 box A,B,5,5,3
190 A=A+JX(1)mul 25;if A>25A=25
```

```

200 if A<-25A=-25
210 B=B+JY(1)mul 25;if B>25B=25
220 if B<-25B=-25
230 if TR(1)goto 3000
240 box A,B,5,5,3
250 goto 180
260 goto 170
501 @(1)=ABS@(1)-1;return
502 @(2)=ABS@(2)-1;return
503 @(3)=ABS@(3)-1;return
504 @(4)=ABS@(4)-1;return
505 @(5)=ABS@(5)-1;return
506 @(6)=ABS@(6)-1;return
507 @(7)=ABS@(7)-1;return
508 @(8)=ABS@(8)-1;return
509 @(9)=ABS@(9)-1;return
600 S=0;for Z=1to 9;S=S+@(Z);next Z
640 if S=8if @(5)=0goto 680
650 if S=0print "YOU LOSE!";goto 1000
660 return
680 &(21)=15;for Z=1to 500;next Z
690 &(21)=0
700 if M>25print "IT TOOK YOU LONG ENOUGH!
710 if M>20if M<26print "HEY! YOU FINALLY GOT IT!
720 if M>15if M<21print "PRETTY GOOD, BUT YOU CAN DO BETTER!
730 if M>10if M<16print "IT TOOK YOU ",#1,M," MOVES";print "THIS TIME.
740 if M>5if M<11print "NOW YOU'RE REALLY GETTING GOOD AT THIS GAME.
750 if M<6print "YOU ARE EITHER AN ACE OR VERY LUCKY!
1000 for Z=1to 3000;next Z;goto 10
2000 box X,Y,23,23,2;box X,Y,23,23,3;if @(U)=0box X,Y,21,21,3;goto 145
2008 NT=3;MU=rnd (8)+48;NT=0
2010 box X,Y,19,19,3;box X,Y,15,15,3;box X,Y,11,11,3;box X,Y,7,7,3;goto 145
3000 if A=0if B=0gosub 505;gosub 502;gosub 504;gosub 506;gosub 508;goto 95
3010 if A=-25if B=-25gosub 501;gosub 502;gosub 504;gosub 505;goto 95
3020 if A=0if B=-25gosub 502;gosub 501;gosub 503;goto 95
3030 if A=25if B=-25gosub 503;gosub 502;gosub 506;gosub 505;goto 95
3040 if A=-25if B=0gosub 504;gosub 501;gosub 507;goto 95
3050 if A=25if B=0gosub 506;gosub 503;gosub 509;goto 95
3060 if A=-25if B=25gosub 507;gosub 504;gosub 505;gosub 508;goto 95
3070 if A=0if B=25gosub 508;gosub 507;gosub 509;goto 95
3080 if A=25if B=25gosub 509;gosub 508;gosub 506;gosub 505;goto 95
5000 print "THE OBJECT OF THE GAME IS to CHANGE THE POSITIONS OF
5010 print "THE BOXES SO THAT THE
5020 print "CENTER IS EMPTY, AND ALL THE OTHER SQUARES ARE
5030 print "FILLED. THIS IS DONE BY
5040 print "MOVING THE FLASHING box
5050 print "WITH THE HAND CONTROLLER, AND PULLING THE TRIGGER.
5060 for Z=1to 5000;next Z;clear ;goto 30
run

```

And here's a complete copy of Enhanced Lunar Lander from W&W Software Tape #7. On the only copy available before, the recording cut off before the program ended.

```
10 :return ;T=0
900 &(23)=255;clear ;BC=0;FC=126;NT=0
1000 for A=-80to 80;T=T+rnd (3)-2;if T<0T=0
1030 line A,-44,0;line A,T-44,1;next A
1060 for A=1to 50;box rnd (160)-80,rnd (88)-44,1,1,1;next A
2000 T=0;V=3;G=0
2035 F=100
2040 X=-75
2050 Y=43;box X,Y,3,3,3;if T box X-(Tmul 2),Y,3,1,3
2063 if U box X,Y-2,1,3,3
2500 box X,Y,3,3,3;if T box X-(Tmul 2),Y,3,1,3
2506 if U box X,Y-2,1,3,3
2508 if F>0U=TR(1);T=JX(1);V=V+JX(1)
2509 if F<1U=0;T=0;F=0
2510 if (U)+(T)&(21)=255
2512 if U=0if T=0&(21)=0
2515 if PX(X,Y)=1goto 6000
2520 G=G+8-20mul U;Y=Y-Gdiv 25;if Y<-41goto 3000
2527 if Y>43print "SHIP IN ORBIT!";&(21)=0;goto 8000
2530 X=X+V;if X>75X=-75
2531 if X<-75X=75
2532 if F<1 U=0
2533 if U F=F-1
2534 if T F=F-1
2535 box X,Y,3,3,3;if U box X,Y-2,1,3,3
2537 if T box X-(Tmul 2),Y,3,1,3
2560 CX=-20;CY=44;print #3,F,#4,Y+41,-Gdiv 10
2590 goto 2500
3000 Y=-41
3006 if (G>40)+(PX(X,Y)=1)goto 5000
3010 G=0;CY=44;CX=-20;print #3,F,#4,Y+41,-Gdiv 10
3030 &(21)=0;box X,Y,3,3,3;box X,Y+3,1,3,3;if U box X,Y-2,1,3,3
3040 box X+2,Y-1,1,1,1;box X-2,Y-1,1,1,1;box X+3,Y-2,1,1,1;box X-3,Y-2,1,1,1
3050 V=0;box X,Y-2,5,1,2
4010 if &(22)=16goto 10
4020 if TR(1)if F>0box X,Y+3,1,3,3;box X,Y-2,7,1,2;box X+2,Y-1,1,1,2;box X-2,Y-1,1,1,2;goto 2500
4025 if &(21)=8goto 7000
4030 goto 4010
5000 BC=92;NT=1;box X,Y,9,9,2;BC=0;for Z=-5to 5
5010 line X,Y,0;line Zmul rnd (5)+X+V,Y+rnd (10),3;MU=49;next Z
5040 NT=0;F=0;&(21)=0;goto 6080
6000 BC=92;NT=1;box X,Y,9,9,2;BC=0;for Z=-5to 5
6020 line X,Y,0;line Zmul rnd (5)+X+V,Y+rnd (10),3;MU=49;next Z
6040 &(21)=0;for Z=-5to 5;line X,Y,0;line Zmul rnd (5)+X+V,Y-rnd (10),3;MU=49;next Z
6080 NT=0;for Z=1to 3000;next Z;goto 10
```

```
7000 for Z=1to 10;box X+8,Y+10-Z,Z,Z,1;next Z
7025 box X+8,Y+10,1,20,1
7026 box X+8,Y+15,7,3,3
7027 box X+13,Y,3,5,1
7030 box X-8,Y+2,9,9,1
7070 for Z=1to 5;box X+11,Y+15,Z,3,2;box X+5,Y+15,Z,3,2;next Z
7080 for Z=1to 7;box X+8,Y+15,Z,3,1;next Z
7110 if TR(1)if F>0goto 4020
7120 goto 7070
8000 for Z=1to 3000;next Z;clear ;print "DOCKING SEQUENCE!
8010 C=1;M=rnd (40)-20;N=rnd (40)-20;A=rnd (5)-3;B=rnd (5)-3;for Z=1to 999;next Z;clear
8030 clear ;box 0,0,C,C,1;box 0,0,3,3,2;box M,0,1,70,1;box 0,N,150,1,1;if F>0D=JX(1);E=JY(1)
8035 CY=44;print #1,F
8037 if M=0if N=0goto 9000
8040 if D F=F-1;&(21)=255
8050 if E F=F-1;&(21)=255
8055 if F<0F=0;D=0;E=0
8060 if D=0if E=0&(21)=0
8080 A=A+D;B=B+E;M=M+A;N=N+B
8110 C=C+2;if C>70X=0;Y=0;goto 6000
8120 goto 8030
9000 print "DOCKING COMPLETE!";goto 5040
run
```

Paul

-----

---

Groups.io Links:

You receive all messages sent to this group.

[View/Reply Online \(#18168\)](#) | [Reply To Group](#) | [Reply To Sender](#) | [Mute This Topic](#) | [New Topic](#)  
[Your Subscription](#) | [Contact Group Owner](#) | [Unsubscribe](#) [ballyalley@hotmail.com]

-----